



Title	マイクロコンピュータを端末として使用する方法
Author(s)	藤井, 博
Citation	大阪大学大型計算機センターニュース. 1980, 38, p. 97-104
Version Type	VoR
URL	https://hdl.handle.net/11094/65457
rights	
Note	

The University of Osaka Institutional Knowledge Archive : OUKA

<https://ir.library.osaka-u.ac.jp/>

The University of Osaka

マイクロコンピュータを端末として使用する方法

大阪大学大型計算機センター

共同利用掛 藤 井 博

最近は、マイクロコンピュータも随分安価になり、各研究室で購入されているのではないのでしょうか？。すでにマイクロコンピュータを購入された方、あるいはTSS端末を購入したいのだからもっと安価な端末はないか、と検討されている方は一読して下さい。


センターニュース№35(1979-11)で、お知らせしました通り遅まきながらマイクロコンピュータ8080系のクロス・アセンブラーが使用できるようになりました。

このたび、クロス・アセンブラーのデバッグのため日本電気社製COMPO BS/80を購入し、このCOMPO BS/80にモデム(RS232C)インターフェースを接続して、音響カプラを通じてTSSを利用できる様に改造しました。(改造はミカサ商事に依頼)

このCOMPO BS/80を例に取り上げ、マイクロコンピュータをTSS端末として使用する方法を述べます。

ハードウェアはCOMPO BS/80 (キャビネットの中には、CPUマザーボード、TK80-BS、TK80-BSキーボード、オートカセットデッキ、オートカセット用SUBボード、インタフェースボード、電源、冷却用ファンが収納されています。)とTK-M20K(拡張用ボードでRAM12KBとI/Oポートが実装されています。)を使用し、音響カプラは、NEC社製DATAX300CBを使用しました。

モデム(RS232C)インターフェースとしては、TK-M20Kに実装されているプログラム汎用非同期送受信LSI(UART)NEC社製 μ PD8251を中心に、バス・ドライバIC、TI社SN75189AN、バス・レシーバIC、ナショセミ社DS1488Nを使用し実現しました。



汎用非同期送受信LSI(8251)

さてハードウェアが出来ただけでは、端末として動きません。このモデム(RS232C)インターフェースをイニシャライズして、端末として動く様にプログラムを作らなければなりません。

モデム(RS232C)インターフェースの中心となっているのは8251型のプログラマブル汎用非同期送受信LSIチップです。このチップは、制御レジスタに次の様な8ビット・データをセットすることによりモード設定をおこないます。

8251 通 信 モ ー ド

ビット	モード	説明
(右) 0 1	クロックレート	$\begin{bmatrix} 0 \\ 1 \end{bmatrix}$ * 1, $\begin{bmatrix} 1 \\ 0 \end{bmatrix}$ * 16, $\begin{bmatrix} 1 \\ 1 \end{bmatrix}$ * 64
2 3	データ長	$\begin{bmatrix} 0 \\ 1 \end{bmatrix}$ 7ビット長, $\begin{bmatrix} 1 \\ 1 \end{bmatrix}$ 8ビット長
4 5	パリティ	$\begin{bmatrix} 0 \\ 0 \end{bmatrix}$ $\begin{bmatrix} 0 \\ 1 \end{bmatrix}$ 無視, $\begin{bmatrix} 1 \\ 0 \end{bmatrix}$ 奇数, $\begin{bmatrix} 1 \\ 1 \end{bmatrix}$ 偶数
6 (左) 7	ストップビット長	$\begin{bmatrix} 1 \\ 0 \end{bmatrix}$ 1ビット, $\begin{bmatrix} 1 \\ 1 \end{bmatrix}$ 2ビット

次に制御レジスタに次のコマンドをセットします。(モード指定後は内部リセットをしない限りコマンド指定となります)

8251 通 信 コ マ ン ド

ビット	コマンド	説明
(右) 0	送信可	1 送信可能にする
1	端末OK	1 端末の用意OK
2	受信可	1 受信可能にする
3	ブ레이크送出	0 通常動作, 1ブ레이크送出
4	エラー・リセット	1 エラーリセットする
5	送信要求	1 送信要求する
6	内部リセット	1 内部リセット, モードセット命令受付可能にする。
(左) 7		

TK-M20K上のI/Oポートはメモリ・マップ化I/Oとなっているためモード指定とコマンド指定は次のようになります。

```

UART    EQU    7C09H    8251 制御レジスタ アドレス
      ⋮
      MVI    A, 01111011B    { 01 = ストップ・ビット 1ビット
                                11 = 偶数パリティ
                                10 = 7ビット
                                11 = 300ボ
      STA    UART

      MVI    A, 00100111B    { 送信可
                                端末OK
                                受信可
                                送信要求
      STA    UART

```

これでモードはストップ・ビット1ビット、偶数パリティ、7ビット・データ、300ボーとなり送信を要求すると同時に、端末を送信可能な状態にすることを意味します。

データをホストに送信するには、次のようになります。

```

UART    EQU    7C09H    8251 制御レジスタ アドレス
COUPL    EQU    7C08H    8251 データレジスタ アドレス
      ⋮
      ⋮
      ⋮
SEND:    LDA    UART    ステータスをアキュムレータに取込む
      ANI    00000001B
      JZ      SEND
      MOV    A, B
      STA    COUPL    ホストに送出

```

ここで最初のLDA命令は、ステータス（状態情報）をアキュムレータに取込む命令です。ステータス（状態情報）は、次の8ビットで表わされます。

8251 ステータス

ビット	状態	説明
(右) 0	送信可	0 送信不可能です、1 送信可能です
1	受信終了	1 データ受信終了です
2	送信終了	1 データ送信終了です
3	パリティ・エラー	1 パリティが合いません
4	オーバーラン・エラー	1 前の文字の入出力が終了しないのに次の文字がきました。
5	フレーミング・エラー	1 ストップ・ビットの個数が正しくありません。
6		
(左) 7		

ホストへ送信可能であれば、アキュムレータには00000001がセットされますので、このステータスをチェックして送信可能であればBレジスタの内容をホストへ送出します。

データをホストより受信するには、次のようになります。

```

UART    EQU    7C09H    8251 制御レジスタ アドレス
COUPL   EQU    7C08H    8251 データ レジスタ アドレス
      ...
RECEV:  LDA     UART      ステータスをアキュムレータに取込む
        ANI     00000010B
        JZ      RECEV
        LDA     COUPL     データをアキュムレータに取込む

```

やはりここでも最初のLDA命令は、ステータス(状態情報)をアキュムレータに取込む命令です。8251はシリアルなデータ・ビットを受信しストップビットを受信して7ビットまたは8ビットのデータが揃うと、ステータスには00000010をセットします。このステータスをチェックして受信終了であれば、データ・レジスタの内容をアキュムレータに取込みます。

マイコンをTSS端末として利用する時にもう一点必要な機能は、割り込みなどの目的でブレーク信号を送出する機能です。これは次の様にします。

```

UART    EQU    7C09H    8251 制御レジスタ アドレス
COUPL   EQU    7C08H    8251 データレジスタ アドレス
      ...
BRAKE:  MVI     D, 07H    } ループカウンタセット
        MVI     E, 0FFH

```

```

ON:      MVI      A, 00101111B
          STA      UART      } ブレーク送出
LOOP:    DCX      D
          MOV      A, D
          CPI      0H
          JZ       OFF
          JMP      LOOP
OFF:     MVI      A, 00100111B
          STA      UART      } ブレーク送出停止

```

ブレーク信号は0レベルが連続500 ms 以上送出された時です。そのためカウンタによって、ブレーク信号送出中はループする様になります。

以上でモデム(RC232C)インターフェースに関するプログラムは終了です。これを基本として、マイクロコンピュータをインテリジェント端末として使用することができます。たとえば、TSS使用時のパスワード自動送出、ホストからマイクロ・コンピュータのカセットテープまたはフロッピーディスクへのファイル転送とその逆のファイル転送、マイクロ・コンピュータに各種のデータ収集装置を接続し直接ホストにデータを転送する等マイクロ・コンピュータの小まわりのきく融通性を活かしてすばらしい端末ができます。

マイクロ・コンピュータは、今後ますますメモリも増え強力になっていくでしょう。プログラムさえうまく作ればいろいろなことができます。この頁がそのための足がかりとなれば幸いです。

最後に、単なる端末としてのプログラムを載せておきます。COMPO BS/80のキーボードには、ブレークキーがありませんので「!」キーで代用しています。

参考文献

石田晴久著 マイクロコンピュータの活かし方 産報出版(1977)

東京大学大型計算機センターニュース

VOL.12 No.1 P.24~P.36 (1980)

UCOM*8A -ASSEMBLY LIST- VERSION 2.0

ST.NO	ADDR	CODE	F	E	LABEL	OP	OPERAND/COMMENT	PAGE 0001
00001		9000				ORG	9000H ;ORIGIN 9000H	
00002		01CA			TVEX4	EQU	01CAH ;CRT CLEAR ORIGIN	
00003		01C4			TVEX1	EQU	01C4H ;1CH DISPLAY ORIGIN	
00004		01BB			KBEX2	EQU	01BBH ;KEY INPUT ORIGIN	
00005		7C09			UART	EQU	7C09H ;8251 MODE SET ORIGIN	
00006		7C08			COUPL	EQU	7C08H ;8251 DATA SET ORIGIN	
00007			*		*COLOR	MACRO	DATA ;COLOR MACRO DEFIN	
00008			*		MVI	A,DATA	;DATA TO A-REG	
00009			*		STA	7DFDH	;SET COLOR DATA	
00010			*		ENDM			
00011	9000	CDCA01			START:	CALL	TVEX4 ;CRT CLEAR	
00012	9003	3EB0				MVI	A,0B0H ;8255 COLOR SET	
00013	9005	32FF7D				STA	7DFFH	
00014	9008	3E70				MVI	A,70H	
00015	900A	32FD7D				STA	7DFDH	
00016	900D	3E02				MVI	A,02H	
00017	900F	32FE7D				STA	7DFEH	
00018	9012	CD7C90				CALL	INIT ;8251 INITIAL	
00019	9015	00				DB	0,0,0	
00020	9016	00						
00021	9017	00						
00022	9018	CD8B01			TERM:	CALL	KBEX2 ;KEY INUT CHECK	
00023	901B	CA3990				JZ	RCV ;IF NO INPUT JUMP RECEV	
00024	901E	47				MOV	B,A ;A-REG TO B-REG	
00025	901F	FE21				CPI	21H ;CHECK KEY INPUT "!"	
00026	9021	CA5E90				JZ	BRAKE ;IF ! JUMP BRAKE	
00027				1		COLOR	05H ;COLOR BLACK AND YELLOW	
00028	9024	3E05		1		MVI	A,DATA ;DATA TO A-REG	
00029	9026	32FD7D		1		STA	7DFDH ;SET COLOR DATA	
00030	9029	78				MOV	A,B ;B-REG TO A-REG	
00031	902A	CDC401				CALL	TVEX1 ;DISPLAY A-REG	
00032	902D	00				DB	0,0,0	
00033	902E	00						
00034	902F	00						
00035	9030	CD8F90				CALL	SEND ;DATA SEND	
00036	9033	C31890				JMP	TERM ;JUMP TERM	
00037	9036	00				DB	0,0,0	
00038	9037	00						
00039	9038	00						
00040	9039	CD9F90			RCV:	CALL	RECEV	
00041	903C	CA1890				JZ	TERM	
00042	903F	FE0A				CPI	0AH	
00043	9041	CA4E90				JZ	\$+13	
00044	9044	FE0D				CPI	0DH	
00045	9046	CA4E90				JZ	\$+8	
00046	9049	FE20				CPI	20H	
00047	904B	FA1890				JM	TERM	
00048	904E	47				MOV	B,A ;A-REG TO B-REG	
00049				1		COLOR	01H ;COLOR BLACK AND GREEN	
00050	904F	3E01		1		MVI	A,DATA ;DATA TO A-REG	
00051	9051	32FD7D		1		STA	7DFDH ;SET COLOR DATA	
00052	9054	78				MOV	A,B ;B-REG TO A-REG	
00053	9055	CDC401				CALL	TVEX1 ;DISPLAY	
00054	9058	C31890				JMP	TERM	
00055	905B	00				DB	0,0,0	

UCOM*8A -ASSEMBLY LIST- VERSION 2.0

ST.NO	ADDR	CODE	F E LABEL	OP	OPERAND/COMMENT	PAGE 0002
00056	905C	00				
00057	905D	00				
00058	905E	1607	BRAKE:	MVI	D,07H ;COUNT SET DE-REG	
00059	9060	1EFF		MVI	E,0FFH	
00060	9062	3E2F	ON:	MVI	A,00101111B ;BRAKE ON	
00061	9064	32097C		STA	UART	
00062	9067	1B	LOOP:	DCX	D ;DECREMENT DE-REG	
00063	9068	7A		MOV	A,D	
00064	9069	FE00		CPI	0H	
00065	906B	CA7190		JZ	OFF	
00066	906E	C36790		JMP	LOOP	
00067	9071	3E27	OFF:	MVI	A,00100111B ;BRAKE OFF	
00068	9073	32097C		STA	UART ;8251 MODE SET	
00069	9076	C31890		JMP	TERM	
00070	9079	00		DB	0,0,0	
00071	907A	00				
00072	907B	00				
00073	907C	3E40	INIT:	MVI	A,01000000B ;RESET MODE SET	
00074	907E	32097C		STA	UART	
00075	9081	3E7B		MVI	A,01111011B ;MODE SET	
00076						
00077						
00078						
00079						
00080	9083	32097C		STA	UART ;8251 MODE SET	
00081	9086	3E27		MVI	A,00100111B ;MODE SET SEND AND RECEIVE	
00082	9088	32097C		STA	UART ;8251 MODE SET	
00083	908B	C9		RET		
00084	908C	00		DB	0,0,0	
00085	908D	00				
00086	908E	00				
00087	908F	3A097C	SEND:	LDA	UART ;8251 STATS SET	
00088	9092	E601		ANI	00000001B ;SEND BUFF CHECK	
00089	9094	CA8F90		JZ	SEND ;IF BUFF FULL JUMP SEND	
00090	9097	78		MOV	A,B ;MOVE B-REG TO A-REG	
00091	9098	32087C		STA	COUPL ;8251 DATA SET	
00092	909B	C9		RET		
00093	909C	00		DB	0,0,0	
00094	909D	00				
00095	909E	00				
00096	909F	3A097C	RECEV:	LDA	UART ;8251 STATS SET	
00097	90A2	E602		ANI	00000010B ;RECEIVE BUFF CHECK	
00098	90A4	CAAD90		JZ	NOREC ;IF NO RECEIVE	
00099	90A7	3A087C		LDA	COUPL ;DATA RECEIVE TO A-REG	
00100	90AA	C600		ADI	0H ;ZERO FLG 1 SET	
00101	90AC	C9		RET		
00102	90AD	97	NOREC:	SUB	A ;ZERO FLG AND A-REG 0 SET	
00103	90AE	C9		RET		
00104				END		

ASSEMBLY COMPLETED ERROR COUNT = 0

*** SYMBOL LIST ***

SYM	ADDR	SYM	ADDR	SYM	ADDR	SYM	ADDR	SYM	ADDR
A	0007	B	0000	BRAKE	905E	C	0001	COUPL	7C08
D	0002	E	0003	H	0004	INIT	907C	KBEX2	01BB
L	0005	LOOP	9067	M	0006	NOREC	90AD	OFF	9071
ON	9062	PSW	0006	RCV	9039	RECEV	909F	SEND	908F
SP	0006	START	9000	TERM	9018	TVEX1	01C4	TVEX4	01CA
UART	7C09								

* * * JOB END * * *